Paper 220-28

Scheduling Time with SAS®: Project Proposal Examples

Barbara B. Okerson, Ph.D., Virginia Health Quality Center, Glen Allen, VA

ABSTRACT

The SAS system provides the ability to develop both basic and presentation-ready tools for project and project proposal time and cost management. While the SAS system includes direct project management tools in SAS/OR using the SAS PROJMAN system, it is possible to create many of the same tools without SAS/OR. This paper focuses on SAS as a project management tool using the procedures available in SAS/GRAPH.

All examples were developed with version 8.2 of SAS executing on a Windows 98 platform. Examples are not platform specific and can be adapted by both beginning and advanced SAS users.

INTRODUCTION

The Virginia Health Quality Center (VHQC), winner of the U.S. Senate Productivity and Quality Award for Progress in Performance Excellence, is a health care quality improvement organization that assists healthcare providers in making meaningful changes in the way care is delivered, and in improving the outcomes of that care, especially for the Medicare community of Virginia. Services provided by the VHQC include assisting healthcare organizations with performance improvement techniques, reviewing health plan denials, statistical consulting and data analysis services, health education, credentials verification, clinical and claims database engineering, health care utilization and quality reviews, and coding/DRG validations.

The VHQC actively seeks collaborations and quality improvement opportunities within the healthcare community. Successful proposals require careful planning and scheduling of time; this process can be greatly assisted by the graphical display of the time involved for the processes required for task completion.

The SAS software system provides all the tools for time scheduling and display through the Base, SAS/GRAPH, SAS/OR and SAS/QC products. The examples that follow illustrate the use of SAS software to create project report timelines, business plans, and project proposals. Data used in these examples are for illustration purposes only; they do not represent actual results or projections.

EXAMPLE ONE

In this example, a time schedule is created for a task within a project. Areas of responsibility are color-coded. This type of schedule is particularly useful because it provides at a glance the same information as well-known project planning software, yet limits the display to the necessary factors while providing the schedule in an easy-to-read, scaleable format.

Example One Graphic:



Example One Code:

```
/* Set Graphics Options*/
goptions reset=all ftext="Arial" htext=2.5
   targetdevice=jpeg device=win;
run;

/* Read in Data Set*/
data process;
   infile "c:\sugi28\example1.dat";
   input Task $40. @43 begin date9. @56 end
       date9. timein $ 43-65 linetype color $;
    time=substr(timein,1,7)||"- "
       ||substr(timein,14,6);
format begin end date9.; nobs=_n_; run;

/*Create format data set for data ordering*/
Data Order;
   set process(rename=(nobs=start time=label));
   fmtname='order'; type="N";
   keep start type fmtname label; run;
proc format cntlin=order;

/*Pass Long Axis Names to Macro variable*/
data _null_;
   set process end=eof;
   call symput('task'||left(_n_),
      "'"||trim(task)||"'");
   if eof then do;
      call symput('total',_n_);
   end;
run;
%macro longtasks;
   %do i=1 %to &total;
      &&task&i
   %end;
%mend longtasks;

/*Create Annotate Data Set to Draw Lines*/
data anno;
   length function color $8 ;
   retain xsys ysys '2' size 10; set process;
   line=linetype; color=color;
   function='move'; x=end+.3; y=nobs; output;
   function='draw';x=begin;y=nobs;output; run;

title1 h=4 c=black "Quality Improvement Report
   Timeline";

/* Set Symbols and Axes*/
symbol1 i=none;
axis1 label=none major=none minor=none style=0 ;
axis2 label=none major=none minor=none style=0
   order="20Feb02"d to "16Mar02"d  value=none
   length=45;
axis3 label = none value=(j=l %longtasks)
   major=none minor=none style=0 ;

/*Plot the Timeline*/
proc gplot annotate=anno;
plot nobs*end/vaxis=axis1 haxis=axis2 lhref=33
   href="01MAR02"d "12MAR02"d vreverse;
plot2 nobs*end/vaxis=axis3 haxis=axis2 vreverse;
format nobs order.;
footnote2  box=1 f=marker c=red h=1.5 'U'
   f='Arial' c=black h=1.5 ' QI            '
   f=marker c=gray h=1.5 'U' f='Arial' c=black
   h=1.5 ' Analytic       '  f=marker c=blue
   h=1.5 'U' f='Arial' c=black h=1.5 ' Other  ';
run;
quit;
```

EXAMPLE TWO

In this example, an example from SAS/OR project management documentation is presented and a similar example created using SAS/GRAPH. A comparison of the SAS code shows that, after initial set-up there is little difference in the programming time required. Additionally, the SAS/GRAPH procedure code combined with the SAS Annotate facility adds flexibility to the graphic.

SAS/OR Example Using Proc CPM and Proc GANTT:



The example above is one used by SAS Institute to illustrate the capabilities of SAS/OR for project management. The CPM procedure produces a schedule and information on resource usage using a model of the project as input. The procedure itself produces no printed output but writes the schedule and the resource usage information to SAS data sets that can be output to the GANTT, Netdraw or other procedures.

The example above uses the GANTT procedure for display. Data elements included in a schedule data set include: schedule times, activities, activity durations, target dates, deadlines, and hierarchy codes. Additional variables can be added by the user. Although not illustrated here and not an efficient solution, the CPM data set could be formatted for use with SAS/GRAPH.

In the example that follows, the same project management principles have been applied to scheduling in the health care quality improvement arena.

Non SAS/OR Example Graphic:



In this graphic, all four axes are used for data display. Categories are color coded by responsible work group. A format variable is created for the axis order. Expenses (right vertical axis) are recorded as cumulative rather than by item. The SAS code follows.

Example Two Code:

```
/* Set Graphics Options*/
goptions reset=all ftext="Arial" htext=4
    targetdevice=jpeg device=win border
    gunit=pct cback=white;
run;

/* Create Data Set*/
data process;
infile "c:\sugi28\example2.dat";
input Task $40. @43 begin date9. @56 end date9.
     cash linetype color $;
     format begin end date9.; nobs=_n_;

/*Create format data set for ordering of data*/
Data Order;
set process(rename=(nobs=start cash=label));
    fmtname='order'; type="N";
    keep start type fmtname label;
run;
proc format cntlin=order; run;

/*Pass Long Axis Names to Macro variable*/
data _null_;
    set process end=eof;
call symput('task'||left(_n_),
    "'"||trim(task)||"'");
    if eof then do;
         call symput('total',_n_);
    end; run;
    %macro longtasks;
         %do i=1 %to &total;
              &&task&i %end;
    %mend longtasks;

/*Pass Cash Value Names to Macro variable*/
data _null_;
    set process end=eof;
call symput('cash'||left(_n_),
    "'"||trim(right(put(cash,dollar10.)))||"'");
    if eof then do;
         call symput('total',_n_); end; run;

/*Pass cash values to axis*/
    %macro longcash;
         %do i=1 %to &total;
         &&cash&i %end;
    %mend longcash;

/*Create Annotate Data Set to Draw Lines*/
data anno;
  length function color $8 ;
  retain xsys ysys '2' size 10; set process;
  line=linetype; color=color;
function='move'; x=end+.3; y=nobs; output;
function='draw'; x=begin; y=nobs; output; run;

/*Create Annotate data set to connect tasks*/
data anno3;
    retain xsys ysys '2' size 1 color 'black';
    set process;
function='move';x=end+.1;y=nobs;  output;
function='draw'; y=nobs+1; output; run;

/*Label top axis*/
data anno2;
    length function text $8;
    retain xsys '2' ysys '3' color 'black' when
         'a' line '1';
    array txt{5} $5. _temporary_ ("01Mar"
         "04Mar" "07Mar" "10Mar" "13Mar" );
    array xx{5} _temporary_ ("01Mar02"d
         "04Mar02"d "07Mar02"d "10Mar02"d
         "13Mar02"d);
```

```
do i=1 to 5;
   x=xx{i};
   text=left(put(txt{i},$5.));
   function='move';  x=x; y=93; output;
   function='label'; x=x; y=93; output; end;
run;

/* Set Title, Symbols and Axes*/
title1 c=black f="Arial Black" h=5
   "Quality Improvement Report Timeline";
symbol1 i=none;
axis1 label = none value=(j=l %longtasks)
major=none
     minor=none style=0  color=black ;
axis2 label =none minor=none style=1 color=black
     major=(number=5) order=("01Mar02"d to
    "13Mar02"d by 3) value=("01Mar" "04Mar"
    "07Mar" "10Mar" "13Mar")length=70;
axis3 label = none color=black major=none
    minor=none style=0 color=black
    value=(j=l %longcash);

/*Plot the Timeline*/
proc gplot annotate=anno data=anno;
   format nobs order. cash dollar10.;
plot nobs*end/vaxis=axis1 haxis=axis2
   anno=anno3 haxis=axis2 vreverse
   href="1Mar02"d to "13MAR02"d  by 3 vref=.8 ;
plot2 nobs*end /vreverse vaxis=axis3 haxis=axis2
   anno=anno2;

/*Set Legend*/
footnote2 h=4 f='Arial' 'Legend:     '
   h=4 f=marker c=red 'U' move=(-.5,+.1) 'U'
   move=(-.5,+0) 'U' f='Arial' c=black h=4 ' QI
            '  f=marker c=gray h=4  'U'
   move=(-.5,+0) 'U' move=(-.5,+0) 'U'
   f='Arial' c=black h=4 ' Analytic           '
   f=marker c=blue h=4 'U' move=(-.5,+0) 'U'
   move=(-.5,+0) 'U'  f='Arial' c=black h=4 '
   Other  ';
run; quit;
```

Although a large amount of code is required for the initial plot of the above, the resulting program can be used as a template for all future plots of this style. In the example above, dates are hard-coded to reduce the amount of code, but can be generated from the data set. Plot lines and additional annotation can easily be added.

EXAMPLE THREE

As the healthcare quality improvement industry, like most industries, is charged with doing more with less, emphasis has moved to tools and models with demonstrated cost-effective successes. As part of this effort, the VHQC has planned the use of collaboratives and partnerships to maximize its healthcare quality improvement dollars. Collaboratives are designed to improve the quality of care that patients receive through system design using proven, evidence based process changes. The healthcare collaboratives planned by the VHQC are based on the break-through series developed by the Institute of Healthcare Improvement (IHI). Partnerships allow the sharing of efforts across organizations with the same quality improvement objectives.

In this example, a one-page business plan is created as part of planned collaboratives for heartcare (heart failure and acute myocardial infarction), surgical infection prevention (SIP), pneumonia, and inpatient discharge documentation (D-Squared). This plan includes an expense plot and bar chart in addition to an event time line.

Example Three Graphic:



Collaborative timelines, projected expense, and projected patient benefit are displayed in a single graphic. The example code follows. Again, a fairly large amount of code is required for the original graphic; however the resulting code provides a template for future project plans.

Example Three Code:

```
/*Set Default Graphics Options*/
goptions targetdevice=bmp ftext=Swiss i=none
gunit=pct htext=3 pct vsize=8.5 hsize=11
    htitle=.0001;

/*Create Data Set*/
data workplan;
infile "c:\working sas\workplan.dat";
input @1 label $20. @23 date mmddyy10.
       Collab $ 34-43 color $ 45-49 space 51-52;
       format date mmddyy10.; run;

/*Create First Annotate Data Set*/
data anno;
    retain function 'label' position '3' hsys
         '3'  ysys '2' xsys '2' size 5;
set workplan; text=label; x=date-30;
    y=space; color=color; run;

/*Create Program to Label Top Axis*/
data dates;
infile "c:\working sas\dates.dat";
input @1 date date7. expense impact;
    year=year(date); qtr=qtr(date);
    qtrdate=yyq(year,qtr);
format expense dollar8. qtrdate yyq5.; run;

/*Create Annotate Data Set to Label Top Axis";
data anno2; set dates(keep=date);
length function text $8 ;
retain xsys '2' ysys '3' color
    'black' when 'a';
    line=1; year=year(date); qtr=qtr(date);
function='move';x=date;y=80;output;
function='label';x=date;y=80;
    text=left(year(date)); output;
function='label'; x=date; y=75;
    text=trim("Q")||left(qtr(date));output;run;

/*Preprocess Text for Top Axis Display*/
proc sort data=anno2 nodupkey;
    by function year qtr text;
proc sort data=anno2;
    by date descending function;
data anno2; set anno2;
if year=lag2(year) and substr(text,1,1) ne "Q"
```

```
     then do; text=" "; end; run;

/*Create Legend*/
data legend;
length text $ 9 function style $ 8 ;
     when ="A"; position ='6';
     infile "c:\working sas\legend.dat";
     input function $ x y color $ xsys $
          ysys $ text $ 9.;
     if function='bar' then do;
          line=3; style='SOLID'; end;
     else do; style='Swiss'; end; run;

/*Modify Goptions for Legend & Create Legend*/
goptions vsize=1 in htext=20 htitle=.0001
     nodisplay;
proc gslide anno=legend;
title " "; note " ";
run;

/*Set Title, Symbols, Pattern and Axes*/
title c=black h=8 f=swissb "QIO Collaborative
     Schedule and Business Plan"; title3 " ";
axis1 order=(1 to 100 by 1) major=none c=black
     minor=none value=none style=1 label=(h=6
          angle=359.999999 "Event" j=c "Date");
axis2 major=none minor=none value=none style=1
     origin=(20 pct,) order=("01Oct02"d to
           "01dec04"d)label=none;
axis3 order=(1 to 100 by 1) major=none c=black
     minor=none value=none style=1 label=none;
axis4 order=(0 to 45000 by 15000) minor=none
     label=(angle=359.99999999 j=r "
     Expense");
axis5 major=none minor=none value=none style=1
     origin=(20 pct,) order=("01Oct02"d to
          "01dec04"d by qtr)label=none;
axis6 order=(0 to 1500 by 500) minor=none
     label=(angle=359.99999999 "Patient"
     j=c "Impact");
symbol1 c=black value=triangle h=4 i=none;
symbol2 value=none; pattern1 value=solid
      color=yellow;

/*Plot the Timeline*/
goptions vsize=3.5 in;
proc gplot data=workplan annotate=anno;
     plot space*date/vaxis=axis1 haxis=axis2
     lhref=35  href="01Jan03"d "01Apr03"d
          "01Jul03"d "01Oct03"d "01Jan04"d
          "01Apr04"d "01Jul04"d "01Oct04"d;
     plot2 space*date/haxis=axis2 vaxis=axis3
          anno=anno2; run;

 /*Plot the Bar Chart*/
 goptions vsize=2 in htext=8 pct htitle=.0001;
 proc gchart data=dates;title;
     vbar qtrdate/sumvar=expense discrete
     coutline=black clipref raxis=axis4
          maxis=axis5 width=20 vref=15000 30000;
     format expense dollar7. ; title " "; run;

 /*Plot the Line Plot*/
 symbol1 value=none i=join c=purple w=3;
 proc gplot data=dates;
     plot impact*date/haxis=axis2 vaxis=axis6
     vref=500 1000 1500; run;

 /*Assign and Display Graphics in Template*/
 goptions display hsize=15 vsize=11;
 proc greplay igout=work.gseg nofs
     tc=saswork.templt template=sugi;
     Treplay 1:gslide 2:gplot1 3:gchart 4:gplot;
 run;
 quit;
```

EXAMPLE FOUR

This example provides a one-page activity proposal for a consulting contract. Categories and activities within categories are displayed on a timeline, with the number of days for each activity also displayed. Like the previous graphics, the code, once developed can serve as a template for future proposals.

Both the generation of new business and the use of outside consultants are essential parts of the healthcare quality improvement industry. In both instances, clear definitions of work to be performed including task and duration contribute to successful endeavors.

Example Four Graphic:



Although the graphic as displayed shows all categories and subcategories, the code allows for alternate display of major categories only. Additional color-coding be added to identify areas of responsibility; proposal section numbers can also be added as left axis values. Again in this example, dates are hard-coded to reduce the amount of code, but can be generated from the data set. The SAS code follows.

Example Four Code:

```
.
/* Set Graphics Options*/
goptions reset=global ftext=swiss
     targetdevice=bmp device=win;

/* Create Data Set*/
data process;
infile 'c:\working sas\sugi28\process.dat';
input task $char22. @24 begin mmddyy10. @35
     end mmddyy10. linetype 46;
     format begin end mmddyy10. task $char22.;
     month1=month(begin); day1=day(begin);
     month2=month(end); day2=day(end);
     days=end-begin;
     dayplot=right(days||"d"); run;

/*Create task order format*/
data addtaskn;
     set process;
     taskord=_n_; format task $char22.; run;
data taskordr;
     set addtaskn(rename=(taskord=start
          task=label));
     fmtname='dataord'; type='N';
     keep fmtname label start type;run;
proc format cntlin=taskordr;

/*Create Annotate Data Set to Draw Lines*/
data anno; length function color $8;
     retain xsys ysys '2';
```

```
set addtaskn; line=linetype;
function='move'; x=begin; y=taskord; output;
function='symbol'; style="marker"; size=1;
     if taskord=1 then do; text='U'; color="red";
          end;
     else if substr(task,1,1)=" " then do;
          text="D"; color="green"; end;
     else do; text='P'; color="blue"; end;
     output;
function='draw'; size=4; x=end; y=taskord;
     output;
function='symbol'; style="marker"; size=1;
if taskord=1 then do; text='U'; color="red";
     end; output; run;

/*Label top axis*/
data dates;
     infile 'c:\working sas\sugi28\dates.dat';
     input @1 date date7.; day=day(date); run;
data anno2; set dates(keep=date);
     length function text $8 ;
     retain xsys '2' ysys '3' color 'black'
          when 'a' size 1; line=1;
     monname=put(date,monname3.);
     function='move'; x=date; y=90; output;
     function='label'; x=date; y=90;
          text=left(monname); output;
     function='label'; x=date; y=88;
          text=left(day(date)); output;run;
proc sort data=anno2 nodupkey;
     by function monname day text;
proc sort data=anno2;
     by date descending function;
data anno2; set anno2;
     if monname=lag2(monname) and
          substr(text,1,2) eq " " then do;
               text=" "; end; run;
data anno2;
     set anno2;
     if lag3(text)=text then text=" "; run;

/*Pass Long Axis Names to Macro variable*/
data _null_; set process end=eof;
   call symput('dayplot'||left(_n_),
          "'"||trim(dayplot)||"'");
   if eof then do; call symput('total',_n_);
   end; run;
%macro days;
   %do i=1 %to &total;
      &&dayplot&i
   %end;
%mend days;

/* Set Titles, Symbols and Axes*/
title j=l box=1 bspace=0 bc=paoy
     "Consulting Proposal"; title3 " ";
axis1  label=none major=none minor=none
     value=(j=l h=1);
axis2 minor=none label=none major=none
     value=none style=0 order=("01Jul02"d to
          "01Jan03"d by month);
axis3 major=none minor=none label=none
     value=(j=l %days);
symbol1 value=none;

/*Plot the Timeline*/
proc gplot annotate=anno data=addtaskn;
plot taskord*end/frame vreverse
     vaxis=axis1 haxis=axis2 caxis=black
     anno=anno2 href="01Aug02"d "01Sep02"d
          "01Oct02"d "01Nov02"d "01Dec02"d
          lhref=34 cframe=white;
plot2 taskord*end/vaxis=axis3 vreverse;
format taskord dataord. end mmddyy10.;
```

```
footnote1 c=black 'Overall  ' c=red font=marker
    height=1 'U' move=(-1,+0)
    draw=(+0,+.5,+1cm,+0) move=(+3.5,+0) "U"
    f=swiss c=black '     Summary  ' c=blue
    font=marker height=1 'P' move=(-1,+0)
    draw=(+0,+.5,+1,+0) move=(+3.5,+0) "P"
    f=swiss c=black ' Activity  ' c=green
    font=marker height=1 'D' move=(-1,+0)
    draw=(-1,+.5,+1cm,+0) move=(+3.8,+0) "D"
    c=black  bc=paoy box=1 bspace=1; run;
```

The shapes used for the line ends are created using the SAS Marker font, but any hardware or software font characters could be substituted. The format variable is used to maintain data set order and vreverse used on the plot statement to display in chronological order. This could also have been accomplished with an original descending sort of the format variable.

CONCLUSION

By using the built-in flexibility of SAS/GRAPH, as well as the capabilities of the SAS Annotate facility, SAS can be used as a robust time schedule and planning software and the output can be used without using additional software for post-processing. Additionally, a complete schedule and plan can be shown on a single page. This is increasingly important as proposal page-limit restrictions continue to tighten. The examples presented serve as a starting point for a continuing adaptation of the SAS system for projecting and scheduling time.

REFERENCES

http://www.sas.com.

http://www.vhqc.org.

Okerson, Barbara B. 2001. "Fun With Timelines: Doing More With SAS/GRAPH® Proc GPLOT," *Proceedings of the Twenty-seventh Annual SAS Users Group International Conference*. SAS Institute Inc. Cary, NC.

SAS Institute Inc. 1999. *SAS/GRAPH Software: Reference, Version 8, Volumes 1 and 2.*

SAS Institute Inc. 2000. *SAS/OR Users Guide: Project Management, Version 8.*

ACKNOWLEDGEMENTS

The analyses upon which this publication is based were performed under contract no. 500-99-VA-03, titled "Utilization and Quality Control Peer Review Organization for the Commonwealth of Virginia," sponsored by the Centers for Medicare & Medicaid Services (CMS), Department of Health and Human Services (DHHS). The content of this publication does not necessarily reflect the view or policies of the DHHS, nor does mention of trade names, commercial products, or organizations imply endorsement by the U.S. Government. The Virginia Health Quality Center (VHQC) assumes full responsibility for the accuracy and completeness of the ideas presented. This

publication is a direct result of the Health Care Quality Improvement Program initiated by CMS, which has encouraged identification of quality improvement projects derived from analysis of patterns of care and therefore required no special funding on the part of VHQC. Ideas and contributions to the VHQC concerning experience in engaging with the issues presented are welcomed.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. For more information contact:

Barbara B. Okerson, Ph.D.
Virginia Health Quality Center
4510 Cox Road, Suite 400
Glen Allen, VA 23060
Phone: 804.289.5320
e-mail: vapro.bokerson@sdps.org

VIRGINIA HEALTH QUALITY CENTER